# Lagrange Multipliers , Maxima, and Minima

## Introduction

This worksheet shows how to maximize or minimize a function $f(x, y)$ subject to a constraint that $q(x, y) = 0$. This is achieved by forming a new function, $g(x, y) = f(x, y) + \mathrm{mu}\, q(x, y)$ , where mu is the Lagrange multiplier.

```
> restart:
```

## Example 1

Calculate the maximum and minimum values of

```
> f:=x^2+y^2;
```

$$f := x^2 + y^2 \tag{2.1}$$

subject to

```
> q:=x^2+y^2+2*x-2*y+1;
```

$$q := x^2 + y^2 + 2x - 2y + 1 \tag{2.2}$$

by forming $g(x, y)$

```
> g:=f+mu*q;
```

$$g := \mu\,(x^2 + y^2 + 2x - 2y + 1) + x^2 + y^2 \tag{2.3}$$

Differentiate $g$ with respect to $x$ and then with respect to y.

```
> exp1:=diff(g,x);
```

$$exp1 := \mu\,(2x + 2) + 2x \tag{2.4}$$

```
> exp2:=diff(g,y);
```

$$exp2 := \mu\,(2y - 2) + 2y \tag{2.5}$$

The values of $x$, $y$ and mu which give the conditional maxima or minima of $f(x, y)$ are found by solving three equations:

$$q(x, y) = 0$$
$$\frac{\partial}{\partial x} f + \mathrm{mu}\left(\frac{\partial}{\partial x} q\right) = 0$$
$$\frac{\partial}{\partial y} f + \mathrm{mu}\left(\frac{\partial}{\partial y} q\right) = 0$$

```
> exp3:=solve({q=0,exp1=0,exp2=0},{x,y,mu});
```

$$exp3 := \{\mu = 2\,RootOf(2\_Z^2 + 4\_Z + 1) + 1, x = RootOf(2\_Z^2 + 4\_Z + 1), y = -RootOf(2\_Z^2 + 4\_Z + 1)\} \tag{2.6}$$

The quadratic roots are real and can be found using

```
> allvalues(exp3);
```

$$\left\{\mu=\sqrt{2}-1, x=-1+\frac{\sqrt{2}}{2}, y=1-\frac{\sqrt{2}}{2}\right\}, \left\{\mu=-1-\sqrt{2}, x=-1-\frac{\sqrt{2}}{2}, y=1+\frac{\sqrt{2}}{2}\right\} \tag{2.7}$$

Check for a *maximum* at the first of these two points by reusing the math output above.

```
> subs({x = -1-1/2*2^(1/2),y = 1+1/2*2^(1/2)},f);
```

$$\left(-1-\frac{\sqrt{2}}{2}\right)^2+\left(1+\frac{\sqrt{2}}{2}\right)^2 \tag{2.8}$$

```
> fmax:=evalf(%);
```

$$fmax := 5.828427124 \tag{2.9}$$

And for a *minimum* at

```
> subs({x = -1+1/2*2^(1/2),y = 1-1/2*2^(1/2)},f):
> fmin:= evalf(%);
```

$$fmin := 0.1715728755 \tag{2.10}$$

Notice that the Lagrange multiplier mu does not need to be evaluated.

Check this result graphically.
Plot the maximal behaviour of f and the circle representing q = 0, which is the unit circle centred at (-1,1).

```
> with(plots):
> plot1 := implicitplot(f = fmax, x=-3..3, y=-3..3, color=BLUE):
> plot3 := implicitplot(q = 0, x=-3..3, y=-3..3, color=RED):
> display({plot1, plot3},title=`Conditional Maximum`,scaling=
  constrained);
```




```
> plot2 := implicitplot(f = fmin, x=-3..3, y=-3..3, color=GREEN):
> display({plot2, plot3}, title=`Conditional Minimum`,scaling=
  constrained);
```




```
> display({plot1, plot2, plot3},title=`Conditional Minimum`,
  scaling=constrained);
```




Note these values are not the maximum or minimum turning values of $f(x, y)$, but the maximum and minimum values of f which lie on the circle $q(x, y)$. Note also the Langrange multiplier mu need not be evaluated.

The constrained maxima and minima can also be visualized using three dimensions:

```
> plot3d({f - fmax,q}, x= -3..0, y= 0..3);
```




```
> plot3d({f - fmin,q}, x = -3..0, y = 0..3);
```




## Example 2

Using the same procedure, calculate the maximum and minimum values of:

```
> restart;
> f := 8*x^2-12*x*y+17*y^2;
```

$$f := 8\,x^2 - 12\,x\,y + 17\,y^2 \tag{3.1}$$

```
> q:=x^2+y^2-1;
```

$$q := x^2 + y^2 - 1 \tag{3.2}$$

```
> g:=f+mu*q;
```

$$g := \mu\,(x^2 + y^2 - 1) + 8\,x^2 - 12\,x\,y + 17\,y^2 \tag{3.3}$$

```
>
  exp1:=diff(g,x);
```

$$exp1 := 2\,\mu\,x + 16\,x - 12\,y \tag{3.4}$$

```
> exp2:=diff(g,y);
```

$$exp2 := 2\,\mu\,y - 12\,x + 34\,y \qquad (3.5)$$

```
> exp4 := solve({q = 0, exp1 = 0, exp2 = 0},{x, y, mu});
```

$$exp4 := \{\mu = -20, x = -RootOf(5\_Z^2 - 1), y = 2\,RootOf(5\_Z^2 - 1)\}, \{\mu = -5, x = 2\,RootOf(5\_Z^2 - 1), y = RootOf(5\_Z^2 - 1)\} \qquad (3.6)$$

```
> allvalues(exp4[1]);
```

$$\left\{\mu = -20, x = -\frac{\sqrt{5}}{5}, y = \frac{2\sqrt{5}}{5}\right\}, \left\{\mu = -20, x = \frac{\sqrt{5}}{5}, y = -\frac{2\sqrt{5}}{5}\right\} \qquad (3.7)$$

$$x = \sqrt{1 - y^2}, x = -\sqrt{1 - y^2}$$

```
> allvalues(exp4[2]);
```

$$\left\{\mu = -5, x = \frac{2\sqrt{5}}{5}, y = \frac{\sqrt{5}}{5}\right\}, \left\{\mu = -5, x = -\frac{2\sqrt{5}}{5}, y = -\frac{\sqrt{5}}{5}\right\} \qquad (3.8)$$

There are two conditional maxima and two minima.

```
> fmin:=subs({x = 2/5*5^(1/2),y = 1/5*5^(1/2)},f);
```

$$fmin := 5 \qquad (3.9)$$

```
> fmax:=subs({x = 1/5*5^(1/2),y = -2/5*5^(1/2)},f);
```

$$fmax := 20 \qquad (3.10)$$

Now plot the functions to confirm these answers - remember the constraint $q$ is in red.

```
> with(plots):
> plot1 := implicitplot(f = fmax, x=-2..2, y=-2..2, color="Blue"):
> plot3 := implicitplot(q = 0, x=-2..2, y=-2..2, color="Red"):
> display({plot1, plot3}, title=`Conditional Maximum`,scaling=
  constrained);
```

*Conditional Maximum*




```
> plot2 := implicitplot(f = fmin, x=-2..2, y=-2..2, color="Green")
  :
> plot3 := implicitplot(q = 0, x=-2..2, y=-2..2, color="Red"):
> display({plot2, plot3}, title=`Conditional Minimum`);
```




```
> display({plot1, plot2, plot3},scaling=constrained);
```




```
> plot3d({f - fmax,q}, x=-2..2, y=-2..2,axes=boxed, title=
  `Conditional Maximum`);
```

*Conditional Maximum*




```
> plot3d({f - fmin,q}, x=-2..2, y=-2..2,axes=boxed, title=
  `Conditional Minimum`);
```

*Conditional Minimum*